
APPENDIX 1:_MPLUS SCRIPT FOR SIMPLE ACE MODEL (FIGURE 1)
title: 

    Univariate ACE model (see Figure 1)

    In a univariate ACE model the variance is 

    v = a**2 + c**2 + e**2

    and the covariances are

    covmz = a**2 + c**2 for the MZ twin pairs

    covdz = 0.5*a**2 + c**2 for the DZ twin pairs

    where

    a = additive genetic effect

    c = shared environment effect

    e = non shared environment effect

    ** = exponentiation symbol.

    Note: Exclamation marks are for comments and ignored by Mplus.

data:

    file is RD.dat;

variable:

    names are famnum zyg

              reject1 reject2 

              dep1 dep2;

    missing are all (99);           !   Specify missing data symbol.

    usevariables are y1 y2;

    grouping is zyg (1=mz 2=dz);    !   Identify the grouping variable.

define:

    !

    !   Define the variables to be analyzed.

    !   y1 : phenotype measure of twin 1

    !   Y2 : phenotype measure of twin 2

    !

    y1=dep1;

    y2=dep2;

analysis: 

    type = general;

model:

    !

    !   The overall model.

    !   We estimate:

    !

    [y1]; [y2];     !   Means

    y1; y2;         !   Variances

    y1 with y2;     !   Covariance

model mz:

    !

    !   The specific model for each group.

    !   Assign names to the parameters.

    !   The parameter names are provided in parentheses.

    !   Same name = equal parameters.

    !

    [y1](m); [y2](m);       !   Means are equal across twins and across MZ and DZ.  

    y1(v); y2(v);           !   Variances are equal across twins and across MZ and DZ.

    y1 with y2(covmz);      !   Covariance in the MZ group.

model dz:

    [y1](m); [y2](m);

    y1(v); y2(v);

    y1 with y2(covdz);      !   Covariance in the DZ group.

model constraint:

    !

    !   Define the model equations.

    !

    new(a* c* e*);      !   The names of the new parameters.

    v = a**2 + c**2 + e**2;

    covmz = a**2 + c**2;

    covdz = 0.5*a**2 + c**2;

    !

    !   Compute the proportion of variance. Note: This command is not necessary for 

    !   model estimation but will provide useful additional information.

    !

    new(pa pc pe);

    pa = (a**2)/(a**2 + c**2 + e**2);

    pc = (c**2)/(a**2 + c**2 + e**2);

    pe = (e**2)/(a**2 + c**2 + e**2);

output:

    cinterval;      !   Request 95% and 99% confidence intervals.
